# $Id$
# Maintainer: Antonio Rojas <arojas@archlinux.org>
# Contributor: Tom Gundersen <teg@jklm.no>
# Contributor: damir <damir@archlinux.org>
# Contributor: Kevin Edmonds <edmondskevin@hotmail.com>

pkgname=libmtp
pkgver=1.1.14
pkgrel=1
pkgdesc="Library implementation of the Media Transfer Protocol"
arch=(i686 x86_64)
url="https://libmtp.sourceforge.net"
license=(LGPL)
depends=(libusb)
source=("https://downloads.sourceforge.net/$pkgname/$pkgname-$pkgver.tar.gz"{,.asc})
sha256sums=('3817d3e296be8f1bc527385585780e70984e8e0d6a0d00349240d67e3df412e8'
            'SKIP')
validpgpkeys=(7C4AFD61D8AAE7570796A5172209D6902F969C95) # Marcus Meißner <meissner@suse.de>

build() {
  cd $pkgname-$pkgver
  ./configure --prefix=/usr --with-udev=/usr/lib/udev
  make
}

check() {
  cd $pkgname-$pkgver
  make -k check
}

package() {
  cd $pkgname-$pkgver
  make DESTDIR="$pkgdir" install
}
